

## G05EAF – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

G05EAF sets up a reference vector for a multivariate Normal distribution with mean vector  $a$  and covariance matrix  $C$ , so that G05EZF may be used to generate pseudo-random vectors.

### 2 Specification

```
SUBROUTINE G05EAF(A, N, C, IC, EPS, R, NR, IFAIL)
  INTEGER          N, IC, NR, IFAIL
  real            A(N), C(IC,N), EPS, R(NR)
```

### 3 Description

When the covariance matrix is non-singular (i.e., strictly positive-definite), the distribution has probability density function

$$f(x) = \sqrt{\frac{|C^{-1}|}{(2\pi)^n}} \exp \{ -(x - a)^T C^{-1} (x - a) \}$$

where  $n$  is the number of dimensions,  $C$  is the covariance matrix,  $a$  is the vector of means and  $x$  is the vector of positions.

Covariance matrices are symmetric and positive semi-definite. Given such a matrix  $C$ , there exists a lower triangular matrix  $L$  such that  $LL^T = C$ .  $L$  is not unique, if  $C$  is singular.

G05EAF decomposes  $C$  to find such an  $L$ . It then stores  $n$ ,  $a$  and  $L$  in the reference vector  $r$  for later use by G05EZF. G05EZF generates a vector  $x$  of independent standard Normal pseudo-random numbers. It then returns the vector  $a + Lx$ , which has the required multivariate Normal distribution.

It should be noted that this routine will work with a singular covariance matrix  $C$ , provided  $C$  is positive semi-definite, despite the fact that the above formula for the probability density function is not valid in that case. Wilkinson [2] should be consulted if further information is required.

### 4 References

- [1] Knuth D E (1981) *The Art of Computer Programming (Volume 2)* Addison–Wesley (2nd Edition)
- [2] Wilkinson J H (1965) *The Algebraic Eigenvalue Problem* Oxford University Press, London

### 5 Parameters

- |    |  |       |
|----|--|-------|
| 1: | A(N) — <i>real</i> array   | Input |
|    | <i>On entry:</i> the vector of means, $a$ , of the distribution.                                 |       |
| 2: | N — INTEGER  | Input |
|    | <i>On entry:</i> the number of dimensions, $n$ , of the distribution.                            |       |
|    | <i>Constraint:</i> N > 0.  |       |
| 3: | C(IC,N) — <i>real</i> array  | Input |
|    | <i>On entry:</i> the covariance matrix of the distribution. Only the upper triangle need be set. |       |

- 4:** IC — INTEGER *Input*  
*On entry:* the first dimension of the array C as declared in the (sub)program from which G05EAF is called.  
*Constraint:*  $IC \geq N$ .
- 5:** EPS — *real* *Input*  
*On entry:* the maximum error in any element of C, relative to the largest element of C.  
*Constraint:*  $0.0 \leq EPS \leq 0.1/N$ .  
 If EPS is less than *machine precision*, *machine precision* is used.
- 6:** R(NR) — *real* array *Output*  
*On exit:* the reference vector for subsequent use by G05EZF.
- 7:** NR — INTEGER *Input*  
*On entry:* the dimension of the array R as declared in the (sub)program from which G05EAF is called.  
*Constraint:*  $NR \geq ((N + 1) \times (N + 2))/2$ .
- 8:** IFAIL — INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry,  $N < 1$ .

IFAIL = 2

On entry,  $NR < ((N + 1) \times (N + 2))/2$ .

IFAIL = 3

On entry,  $IC < N$ .

IFAIL = 4

On entry,  $EPS < 0.0$ ,  
 or  $EPS > 0.1/N$ .

IFAIL = 5

The covariance matrix  $C$  is not positive semi-definite to accuracy EPS.

## 7 Accuracy

The maximum absolute error in  $LL^T$ , and hence in the covariance matrix of the resulting vectors, is less than  $(n \times \max(EPS, \epsilon) + (n + 3)\epsilon/2)$  times the maximum element of  $C$ , where  $\epsilon$  is the *machine precision*. Under normal circumstances, the above will be small compared to sampling error.

## 8 Further Comments

The time taken by the routine is of order  $n^3$ .

It is recommended that the diagonal elements of  $C$  should not differ too widely in order of magnitude. This may be achieved by scaling the variables if necessary. The actual matrix decomposed is  $C+E = LL^T$ , where  $E$  is a diagonal matrix with small positive diagonal elements. This ensures that, even when  $C$  is singular, or nearly singular, the Cholesky Factor  $L$  corresponds to a positive-definite covariance matrix that agrees with  $C$  within a tolerance determined by EPS.

## 9 Example

The example program prints five pseudo-random observations from a bivariate Normal distribution with means vector

$$\begin{bmatrix} 1.0 \\ 2.0 \end{bmatrix}$$

and covariance matrix

$$\begin{bmatrix} 2.0 & 1.0 \\ 1.0 & 3.0 \end{bmatrix},$$

generated by G05EAF and G05EZF after initialisation by G05CBF.

### 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      G05EAF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          N, IC, NR
      PARAMETER        (N=2, IC=N, NR=(N+1)*(N+2)/2)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, J
*      .. Local Arrays ..
      real             A(N), C(IC,N), R(NR), Z(N)
*      .. External Subroutines ..
      EXTERNAL         G05CBF, G05EAF, G05EZF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G05EAF Example Program Results'
      WRITE (NOUT,*)
      A(1) = 1.0e0
      A(2) = 2.0e0
      C(1,1) = 2.0e0
      C(2,2) = 3.0e0
      C(1,2) = 1.0e0
      C(2,1) = 1.0e0
      CALL G05CBF(0)
      IFAIL = 0

*
      CALL G05EAF(A,N,C,IC,0.01e0,R,NR,IFAIL)
*
      DO 20 I = 1, 5
          IFAIL = 0
          CALL G05EZF(Z,N,R,NR,IFAIL)
          WRITE (NOUT,99999) (Z(J),J=1,N)
      20 CONTINUE

```

```
      STOP
*
99999 FORMAT (1X,2F10.4)
      END
```

## 9.2 Program Data

None.

## 9.3 Program Results

G05EAF Example Program Results

1.7697	4.4481
3.2678	3.0583
3.1769	2.3651
-0.1055	1.8395
1.2933	-0.1850

---